  

	******************************************************************
	**
	**
	**		NAME:		RACHEL BRULE
	**		DATE: 		13 September, 2018
	**		PROJECT: 	"Reform, Representation & Resistance"
	**
	**		DETAILS: 	This file analyzes the impact of a father's treatment
	**					by reservations for households surveyed in REDS 2006/9.
	**					Output for article tables is provided here.
	**					All individual- and village-level identifiers have been
	**					randomly generated to protect human subjects' privacy.
	**
	**				
	**		Version: 	Stata SE 14
	**
	******************************************************************
	




clear
clear mata
clear matrix


set more off
set maxvar 32767
set matsize 10000
set max_memory .
		

*==============================================================================*
****** Specify "mainpath", the location where this replication package is saved.	
****** Then uncomment the remaining global variables to add your file locations.
	

global mainpath 	""


*global datapath 	"$mainpath/Data/source_datasets"
*global dopath		"$mainpath/Do_files/Analysis"
*global analysispath "$mainpath/Data/analysis_datasets"
*global outputpath 	"$mainpath/output"
*==============================================================================*
	

* Set directory for main data
*----------------------------	

****** Uncomment below if using a global variable here:
*cd "$analysispath"



* PREPARATION: Generate ID for genetically matched data
*------------------------------------------------------
use "MatchedData_deidentified.dta", clear

generate genMatchId = 1

save "MatchedData_deidentified_1.dta", replace





* MAIN ANALYSIS: Use main dataset to build core analysis
*-------------------------------------------------------

use "analysis_individuals_deidentified.dta", clear


* Create controls for regression analysis
*----------------------------------------
global controls castedum_1 castedum_2 castedum_3 castedum_4 childT ///
					  wealthy west sibF sibM

				  
* MERGE WITH GENETICALLY MATCHED SUBSAMPLE	
*-----------------------------------------
merge 1:1 villageid interviewno mid using "MatchedData_deidentified_1.dta", ///
		  keepusing(genMatchId)


* Create ID for those who are not included in the genetically matched subsample
replace genMatchId = 0 if genMatchId == .

drop _merge
			

* Change directory for output
*----------------------------

****** Uncomment below if using a global variable
*cd "$outputpath"	


	
*-------------------------------------------------------------------------------
* MAIN TABLES & IN-TEXT ANALYSIS
*-------------------------------------------------------------------------------
	
	
*-------------------------------------------------------------------------------
*  Descriptive statistics * 
*-------------------------------------------------------------------------------
	
* Average age at marriage	
sum marriageage

* NOTE: mean  = 18.07485
* CITE: on pg. 13 (excluding title and abstract pages)


			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**


*-------------------------------------------------------------------------------
*  Labeling variables * 
*-------------------------------------------------------------------------------

label variable dres1 "Father died post reservations"
label variable da_diedpost "Father died post reform"
label variable da_postXdres1 "Father died post reform & reservations"
label variable treat_age120 "Aged < 20 at reform"
label variable dtreat_age120Xpost "Aged < 20 at reform * Father died post reform"
label variable treat_age120Xdres1 "Aged < 20 at reform * Father died post reservations"
label variable dtreat_age120XpostXdres1 "Aged < 20 at reform * Father died post reform & reservations"

*-------------------------------------------------------------------------------
* Table 1: Reservation's impact on women's inheritance, Columns 1-4
*-------------------------------------------------------------------------------

preserve 

eststo clear

		 
* Set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956)


* column 1 - no controls, state & year FE, state trends, landed Hindus only				
			
eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)


* column 2 - controls, state & year FE, state trends, landed Hindus only				
			
eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)


* column 3 - controls, St & Yr FE & St trends, Target - non-random implementers
eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), cl(villageid)


* Column 4 - controls, state & year FE & state trends, LH - (non-random impl + late).

eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), cl(villageid)

			
restore

		
esttab  using "Table1_Column1to4.rtf", b(2) se(2) replace label ///
		title(Table 1: Reservation's Impact on Women's Inheritance) ///
		mtitle("Target" "Target" "Target - NR" ///
		"Target-NR-late") ///
		keep(dres1 da_diedpost da_postXdres1) ///
		order(dres1 da_diedpost da_postXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell

esttab  using "Table1_Column1to4_ref.rtf", b(2) p(3) replace label ///
		title(Table 1: Reservation's Impact on Women's Inheritance) ///
		mtitle("Target" "Target" "Target - NR" ///
		"Target-NR-late") ///
		keep(dres1 da_diedpost da_postXdres1) ///
		order(dres1 da_diedpost da_postXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell	

eststo clear
		
	
	
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

			
*------------------------------------------------------------
* Table 1 - DESCRIPTIVE STATISTICS for results interpretation
*------------------------------------------------------------


* Calculate baseline rate of inheritance: control group
*------------------------------------------------------
* Target women, excluding states w/non-random reservations implementation

sum Tiland_any if (fem == 1 & t < 0 & birthyr > 1956 & landed == 1 & ///
				   Hindu1 == 1 & nr_short != 1 & dres1 == 0 & da_diedpost == 0)
* Mean = .1030011 
* Cited on pg. 18, excl. title and acknowledgements page	
		
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**


			
*-------------------------------------------------------------------
* Table 1 - Reservation's impact on women's inheritance, Columns 5-8
*-------------------------------------------------------------------
		
****** Set calculations for Net Effects
		
* Effect of father death post reform 
*-----------------------------------

* DR1TH1 - Impact: father death post- vs. pre-reform on pr(inherit)|aged less than 20 
* at reform + father death post-reservations
global DR1TH1 "da_diedpost + da_postXdres1 + dtreat_age120Xpost + dtreat_age120XpostXdres1"

* DR0TH1 - Impact: father death post- vs. pre-reform on pr(inherit)|aged less than 20 
* at reform + father death pre-reservations
global DR0TH1 "da_diedpost +  dtreat_age120Xpost"


* Set titles for Excel file
*--------------------------

matrix baseline = J(1,2,-9)
matrix colnames baseline = "Mean" "Standard Deviation" 
matrix rownames baseline =  "Hindu Landed Control"
								
matrix output_hightr = J(2,3,-9)
matrix colnames output_hightr = "Estimate" "SE" "Tscore"
matrix rownames output_hightr = "Change: no ref to ref if reserv" ///
								"Change: no ref to ref if no reserv" 					
										
							
preserve 

eststo clear


* Set appropriate sample
*-----------------------
keep if (fem == 1 & t < 0 & birthyr > 1956)

putexcel set "Table1_Column7_net.xls", replace


* column 5 - only landed hindus, no controls, state & year FE, state trends			
			
eststo: quietly reg Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)


* column 6 - only landed hindus, controls, state & year FE, state trends			
			
eststo: quietly reg Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)


* column 7 - only landed hindus, controls, state & year FE, state tr - not rand,			
			
eststo: quietly reg Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), cl(villageid)

est store T2_LH_NR	
		

* Landed, Hindu control mean
*---------------------------
sum Tiland_any if (fem == 1 & birthyr > 1956 & t < 0 & Hindu1 == 1 & landed == 1 & ///
					dres1 == 0 & da_diedpost == 0 & treat_age120 == 0)
matrix baseline[1,1] = r(mean)
matrix baseline[1,2] = r(sd) 

putexcel F5 = matrix(baseline), names

					
* Main effects
*-------------

* Father dies post-reform, as reservations vary
lincom $DR1TH1

matrix output_hightr[1,1] = r(estimate)
matrix output_hightr[1,2] = r(se)
matrix output_hightr[1,3] = r(estimate)/r(se)

lincom $DR0TH1

matrix output_hightr[2,1] = r(estimate)
matrix output_hightr[2,2] = r(se)
matrix output_hightr[2,3] = r(estimate)/r(se)

putexcel A1 = ("Table 1, col 5-8")
putexcel A3 = ("col 7 - controls, FE, trends, Landed Hindus - Non-random")
putexcel A4 = ("Aged less than 20 at reform") 
putexcel A5 = matrix(output_hightr), names
* Note: Net effect for paternal death post- vs. pre-reform if reservations =
* 		0.103338815 (see Table1_Column7_net.xls, cell C6).
*		Cited on pg. 22, including title and abstract pages.


* column 8 - only landed hindus, controls, state & year FE, state tr - not rand-late,			
			
eststo: quietly reg Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), cl(villageid)

esttab  using "Table1_Column5to8.rtf", b(2) se(2) replace ///
		label title(Table 1: Reservation's Impact on Women's Inheritance) ///
		mtitle("Target" "Target" "Target-NR" "Target-NR-late") ///
		keep(treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ) ///
		order(dres1 da_diedpost da_postXdres1 ///
			treat_age120 dtreat_age120Xpost treat_age120Xdres1  ///
			dtreat_age120XpostXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
				star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell
		
esttab  using "Table1_Column5to8_ref.rtf", b(2) p(3) replace ///
		label title(Table 1: Reservation's Impact on Women's Inheritance) ///
		mtitle("Target" "Target" "Target-NR" "Target-NR-late") ///
		keep(treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ) ///
		order(dres1 da_diedpost da_postXdres1 ///
			treat_age120 dtreat_age120Xpost treat_age120Xdres1  ///
			dtreat_age120XpostXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell
	
restore
eststo clear


			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
		
		
*------------------------------------------------------------
* Table 2 - DESCRIPTIVE STATISTICS for results interpretation
*------------------------------------------------------------


* Calculate baseline rate of participation (0-1): control group
*--------------------------------------------------------------
* Men, all states
sum attendance if (fem == 0 & reside == 1 & age_now > 17 & birthyr != . & ///
				   vwreserv3 == 0)

* Mean = .405825 (rounded to 41 percent)
* Cited in Table 2 note. 	


* Women, all states
sum attendance if (fem == 1 & reside == 1 & age_now > 17 & birthyr != . &  ///
				   vwreserv3 == 0)

* Mean = .1139096 (rounded to 11 percent)
* Cited in Table 2 note.



* Calculate baseline evaluation of Pradhan effectiveness (1-3): control group
*----------------------------------------------------------------------------

* Men, all states
sum resolvesoc1 if (fem == 0 & reside == 1 & age_now > 17 & birthyr != . & ///
				   vwreserv3 == 0)

* Mean = 2.016753 (rounded to 2.02)
* Cited in Table 2 note. 	


* Women, all states
sum resolvesoc1 if (fem == 1 & reside == 1 & age_now > 17 & birthyr != . &  ///
				   vwreserv3 == 0)

* Mean =  1.973897 (rounded to 1.97)
* Cited in Table 2 note.



			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
					

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			
				
*-------------------------------------------------------------------------------
* Table 2, Part I: Reservations' Impact: Women's Participation in Gram Sabha,
*				   COLUMNS 1-4
*-------------------------------------------------------------------------------
	
			
preserve

* keep adult HH residents only (all survey respondents who can participate)
*--------------------------------------------------------------------------

keep if (reside == 1 & age_now > 17 & birthyr != .)



* Regression 4
*-------------

char fem [omit] 0

eststo clear				
		


* Col. 1 - All, without controls
eststo: quietly xi: reg attendance fem vwreserv3  fem_vwreserv3  ///
		vdum* yrdum*, cl(villageid) 
		
* Col. 2 - All, controls
eststo: quietly xi: reg attendance fem vwreserv3 fem_vwreserv3  ///
		vdum* yrdum* $controls, cl(villageid) 
				
* Col. 3. - Excluding non-random, with controls	
eststo: quietly xi: reg attendance fem vwreserv3 ///
		fem_vwreserv3 ///
		vdum* yrdum* $controls if (nr_short != 1), cl(villageid)
				
* Col. 4. - Excluding non-random+late, with controls	
eststo: quietly xi: reg attendance fem vwreserv3 ///
		fem_vwreserv3 ///
		vdum* yrdum* $controls if (nr_short != 1 & late != 1), cl(villageid)		
				

restore
		
		
esttab using "Table2_Column1to4.rtf", b(2) se(2) replace label ///
		title(Reservation's Impact on Women's Participation in Gram Sabha and ///
		Pradhan's Effectiveness) ///
		mtitle("All" "All" "All"  "All-NR" "All-NR-late") ///
		keep ( fem vwreserv3 ///
		fem_vwreserv3) ///
		indicate("Controls = $controls" ///
		 "Village FE = vdum*" ///
		 "Birth year FE = yrdum*") ///
		varwidth(15) modelwidth(10) ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) onecell
		
esttab using "Table2_Column1to4_ref.rtf", b(2) p(3) replace label ///
		title(Reservation's Impact on Women's Participation in Gram Sabha and ///
		Pradhan's Effectiveness) ///
		mtitle("All" "All" "All"  "All-NR" "All-NR-late") ///
		keep ( fem vwreserv3 ///
		fem_vwreserv3) ///
		indicate("Controls = $controls" ///
		 "Village FE = vdum*" ///
		 "Birth year FE = yrdum*") ///
		varwidth(15) modelwidth(10) ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) onecell		
		
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**



			
*-------------------------------------------------------------------------------
* Table 2, Part II: Reservations' Impact, Evaluation of Pradhan Effectiveness, 
*				    COLUMNS 5-8
*-------------------------------------------------------------------------------
			
			
preserve

* keep adult HH residents only (all survey respondents who can participate)
*--------------------------------------------------------------------------

keep if (reside == 1 & age_now > 17 & birthyr != .)

		
* regression 5
*-------------

char vwreserv [omit] 0
eststo clear		


* Col 5: All - no controls, no trends		
eststo: quietly xi: reg resolvesoc1 fem  vwreserv3  ///
		fem_vwreserv3   ////
		vdum* yrdum*, cl(villageid) 
		
	
* Col 6: All - controls	
eststo: quietly xi: reg resolvesoc1 fem  vwreserv3  ///
		fem_vwreserv3   ///
		vdum* yrdum* $controls, cl(villageid) 
				
* Col 7: Excluding non-random - controls
eststo: quietly xi: reg resolvesoc1 fem vwreserv3 ///
		fem_vwreserv3  ///
		vdum* yrdum* $controls if (nr_short != 1), cl(villageid)
 				
		
* Col 8: Excluding non-random & late - controls
eststo: quietly xi: reg resolvesoc1 fem vwreserv3 ///
		fem_vwreserv3 ///
		vdum* yrdum* $controls if (nr_short != 1 & late != 1), cl(villageid)		
		
restore


esttab using "Table2_Column5to8.rtf", b(2) se(2) replace label  ///
		title(Reservation's Impact on Women's Participation in Gram Sabha and ///
		Pradhan's Effectiveness) ///
		mtitle("All"  "All" "All-NR" "All-NR-late" ///
		"All"  "All" "All-NR" "All-NR-late") ///
		keep (fem vwreserv3 fem_vwreserv3) ///
		indicate("Controls = $controls" ///
		"Village FE = vdum*" "Birth year FE = yrdum*") ///
		varwidth(15) modelwidth(10) ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) onecell
		
esttab using "Table2_Column5to8_ref.rtf", b(2) p(3) replace label  ///
		title(Reservation's \& Inheritance's Impact on Evaluation ///
		of Pradhan's Degree of Effectiveness (1-3), `tmp') ///
		mtitle("All"  "All" "All-NR" "All-NR-late" ///
		"All"  "All" "All-NR" "All-NR-late") ///
		keep (fem vwreserv3 fem_vwreserv3) ///
		indicate("Controls = $controls" ///
		"Village FE = vdum*" "Birth year FE = yrdum*") ///
		varwidth(15) modelwidth(10) ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) onecell		
		

		
		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			
			
*-------------------------------------------------------------------------------
* TABLE 3: RESERVATIONS, REFORM, & AGE IMPACT ON DOWRY (GIVEN)
*-------------------------------------------------------------------------------
			
			
preserve 

* set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956)

 
eststo clear
			
* column 1 - no controls, state & year FE, state trends, landed Hindus			
eststo: quietly reg any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ///
			stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)
			

* column 2 - controls, state & year FE, state trends, landed Hindus
eststo: quietly reg any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)

* column 3 - C2 excluding non-random implementers
eststo: quietly reg any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), cl(villageid)
						
* column 4 - C2 excluding non-random & late implementers

eststo: quietly reg any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), ///
			cl(villageid)		
			
restore


esttab using "Table3.rtf", replace se ///
			title(Reservations' Impact on Women's Dowry) ///
			mtitle("Target" "Target" "Target-NR" "Target-NR-late") ///
			cells(b(star fmt(2)) ///
			se(par fmt(2))) keep(treat_age120 dres1 ///
			da_diedpost da_postXdres1 dtreat_age120Xpost treat_age120Xdres1 ///
			dtreat_age120XpostXdres1 ) ///
			order(treat_age120 dres1 ///
			da_diedpost da_postXdres1 dtreat_age120Xpost treat_age120Xdres1 ///
			dtreat_age120XpostXdres1) ///
			indicate("Controls = $controls" ///
			"State FE = stdum*" "Cohort FE = yrdum*" "State trends = sttrend*") ///
			label 	star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
			varwidth(15) modelwidth(10) ///
			stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) onecell
			
esttab using "Table3_ref.rtf", replace b(2) p(3) ///
			title(Reservations' Impact on Women's Dowry) ///
			mtitle("Target" "Target" "Target-NR" "Target-NR-late") ///
			cells(b(star fmt(2)) ///
			se(par fmt(2))) keep(treat_age120 dres1 da_diedpost ///
			dtreat_age120Xpost da_postXdres1 treat_age120Xdres1 ///
			dtreat_age120XpostXdres1) ///
			order(treat_age120 dres1 da_diedpost da_postXdres1 ///
			dtreat_age120Xpost treat_age120Xdres1 dtreat_age120XpostXdres1) ///
			indicate("Controls = $controls" ///
			"State FE = stdum*" "Cohort FE = yrdum*" "State trends = sttrend*") ///
			label 	star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
			varwidth(15) modelwidth(10) ///
			stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) onecell		


eststo clear			
	
		  

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

			
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

	
	
*-------------------------------------------------------------------------------
* APPENDIX TABLES
*-------------------------------------------------------------------------------


*-------------------------------------------------------------------------------
* Appendix Table A1: Descriptive Statistics: Villages without vs. w/Reservations
*-------------------------------------------------------------------------------

	
preserve

eststo clear


* descriptive statistics: by village averages	
*--------------------------------------------

collapse total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 vwreserv, ///
	by(villageid)
	
estpost tabstat  total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99, ///
		 stats (mean sem) columns (stats)	
est store A		

estpost tabstat  total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 ///
		 if vwreserv==1, ///
		 stats (mean sem) columns (stats)	
est store B		 

			
estpost tabstat total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 ///
		 if vwreserv==0, ///
		 stats (mean sem) columns (stats)	
est store C		
		
* ttests
		
estpost ttest total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99, ///
		 by(vwreserv) unequal
		
est store D
			
restore


label variable total_pop "District population, 1991 census"
label variable fem_ratio "% women in the subdistrict (tehsil) population, 1991 census"
label variable vpanch_pop1 "Village population: first panchayat period"
label variable vpanch_memb1 "Number of panchayat members: first panchayat"
label variable vpscmemb_gp1 "% SC panchayat members: first panchayat period"
label variable vpstmemb_gp1 "% ST panchayat members: first panchayat period"
label variable vpobcmemb_gp1 "% OBC panchayat members: first panchayat period"
label variable Vphind "% Hindu in village population currently"
label variable Vpmuslim "% Muslim in village population currently"
label variable Vpsc "% SCs in village population currently"
label variable Vpst "% STs in village population currently"
label variable Vpobc "% OBCs in village population currently" 	
label variable vprmargin "% own less than 2 acres of land in village population currently"
label variable vprowner "% own land in village population currently"
label variable vurlandpr "Average price: unirrigated land currently (Rs.)"
label variable vreslandpr "Average price: residential land currently (Rs.)"
label variable vdrought_99 "% villages experienced drought, 1999"
label variable vflood_99 "% villages experienced flood, 1999"
label variable vpest_99 "% villages experienced pests, 1999"	

* store output
*-------------
esttab A B C using "TableA1_descriptive_statistics.rtf", replace label ///
		title(Descriptive statistics: Villages without vs. with reservations) ///
		mtitle("All Villages" "With Reservations" "Without Reservations") ///
		note("Source: Rural Economic and Demographic Survey, 2006" ///
		"Village level means are provided.") ///
		cells(mean(fmt(2))) main(mean) aux(sem)
			
esttab D using "TableA1_ttest.rtf", label replace main(b) ///
		mtitle("Difference of means: No Reservations - Reservations (t-score)") ///
		b(2) t(2) nostar

		
eststo clear
		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
		


*-------------------------------------------------------------------------------
* Appendix Table A2
*-------------------------------------------------------------------------------


preserve


* Generate appropriate sample
*----------------------------
drop if (t >= 0 | birthyr < 1957)

keep if (femalec == 100 & landed == 1)

keep if (nr_short != 1 & late != 1)


* descriptive statistics: individual respondents	
*-----------------------------------------------
estpost tabstat educmhi educhi top20land Tlandhp Hindu1 LC ///
		 age2 educ prop_sis Tiland_any Tiland, ///
		 stats (mean sd) columns (stats)	
		 
est store A1


estpost tabstat educmhi educhi top20land Tlandhp Hindu1 LC ///
		 age2 educ prop_sis Tiland_any Tiland if dres1 == 1, ///
		 stats (mean sd) columns (stats)	
		 
est store B1

			
estpost tabstat educmhi educhi top20land Tlandhp Hindu1 LC ///
		 age2 educ prop_sis Tiland_any Tiland if dres1 == 0, ///
		 stats (mean sd) columns (stats)	
		 
est store C1	

* ttests
	
estpost ttest educmhi educhi top20land Tlandhp Hindu1 LC ///
		 age2 educ prop_sis Tiland_any Tiland, by(dres1) unequal
		
est store D1


* label variables

label variable Tiland "Total land inherited"

* Output:
*--------

esttab A1 B1 C1 using "TableA2_descriptive_statistics.rtf", replace ///
		title(Balance Test: Father's Death Post vs. Pre-Reservations) ///
		mtitle("All States" "Father Dies Post Reservations" "Pre Reservations")  ///
		label cells(mean(fmt(2))) main(mean) aux(sd)
		
	
esttab D1 using "TableA2_ttest.rtf", replace label ///
		title(Balance Test: Father's Death Pre vs. Post Reservations, T Test) ///
		mtitle("Difference (t-score)") nonum collabels(none) gaps	

	

restore		



	
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

		
			

*-------------------------------------------------------------------------------
* Measurement checks *
*-------------------------------------------------------------------------------


*-------------------------------------------------------------------------------
* Appendix Table A3: Correlation between reservations and pradhan's gender
*-------------------------------------------------------------------------------


preserve
	
* simplify variable names
*------------------------	
rename vp1_fem vp_fem1
rename vp2_fem vp_fem2
rename vp3_fem vp_fem3
	
	
* create village-level observations
*----------------------------------
collapse vp_fem* velect* vwreserv1 vwreserv2 vwreserv3 nr_short late, by(villageid) 	


* Reshape data so we have village - election cycle observations
*--------------------------------------------------------------	
reshape long velect vp_fem vwreserv, i(villageid) j(period)
	
label var vp_fem "Pradhan is female"
label var vwreserv "Pradhan seat is reserved for women"	
	
*Check to what extent reservations predict having a female pradhan
*-----------------------------------------------------------------
eststo clear

eststo: reg vp_fem vwreserv, robust	

eststo: reg vp_fem vwreserv if velect >= 1993, robust

eststo: reg vp_fem vwreserv if (velect >= 1993 & nr_short != 1), robust

eststo: reg vp_fem vwreserv if (velect >= 1993 & nr_short != 1 & late != 1), robust
	
esttab using "TableA3.rtf", b(2) se(2) replace label ///
		title(Reservation's impact on Pradhan's gender) ///
		mtitle("All" "All" "All-NR" "All-NR-late") /// 
		keep(vwreserv) ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) 
				
restore
	
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

		

*-------------------------------------------------------------------------------
* Appendix Table A4: Father death post-res & pradhan gender at death correlation
*-------------------------------------------------------------------------------
	
preserve	
	
* Construct variable: pradhan female at the time of father's death
*-----------------------------------------------------------------
* Father died in election period 1 (first)
gen da_diedp1 = (velect1 <= da_deathyr & da_deathyr < (velect2))
replace da_diedp1 = . if (da_deathyr == . | velect1 == .) 
la var da_diedp1 "Father died in prior-to-prior panchayat period"

* Father died in election period 2 (second)
gen da_diedp2 = (velect2 <= da_deathyr & da_deathyr < (velect3))
replace da_diedp2 = . if (da_deathyr == . | velect2 == .)
la var da_diedp2 "Father died in prior panchayat period"	

* Father died in election period 3 (third)
gen da_diedp3 = (velect3 <= da_deathyr & da_deathyr < (velect3 + 5))
replace da_diedp3 = . if (da_deathyr == . | velect3 == .)	
la var da_diedp3 "Father died in current panchayat period"

* Aggregate information
gen prfem_fdeath = . 
foreach num of numlist 1 2 3 {
	replace prfem_fdeath = 0 if (da_diedp`num' == 1 & vp`num'_fem == 0)
	replace prfem_fdeath = 1 if (da_diedp`num' == 1 & vp`num'_fem == 1 & ///
								 prfem_fdeath == .)
} 

la var prfem_fdeath "Pradhan is female at the time of father's death"

	
*Check to what extent reservations predict having a female pradhan
*-----------------------------------------------------------------
eststo clear

eststo: quietly reg prfem_fdeath dres1, cl(villageid)	

eststo: quietly reg prfem_fdeath dres1 if (da_deathyr >= 1993), cl(villageid)

eststo: quietly reg prfem_fdeath dres1 $controls if (da_deathyr >= 1993), cl(villageid)

eststo: quietly reg prfem_fdeath dres1 $controls yrdum* stdum* sttrend* ///
			if (da_deathyr >= 1993), cl(villageid)
			
eststo: quietly reg prfem_fdeath dres1 $controls yrdum* stdum* sttrend* ///
			if (da_deathyr >= 1993 & nr_short != 1), cl(villageid)
			
eststo: quietly reg prfem_fdeath dres1 $controls yrdum* stdum* sttrend* ///
			if (da_deathyr >= 1993 & nr_short != 1 & late != 1), cl(villageid)

	
esttab using "TableA4.rtf", b(2) se(2) replace label ///
		title(Impact of father's death post-reservations on Pradhan's gender at ///
		father's death) mtitle("All" "All" "All" "All" "All-NR" "All-NR-late") /// 
		keep(dres1) order(dres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) 	

	
restore	
	
			
		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

			
			
	
*-------------------------------------------------------------------------------
* Appendix Table A6: Descriptive Statistics, Individuals
*-------------------------------------------------------------------------------


* label variables

label variable Tiland "Total land inherited (acres)"



preserve


* set sample
*-----------
drop if (t >= 0 | birthyr < 1957)

estpost tabstat Hindu1 Tiland_any Tiland age2 educ prop_sis educmhi educhi ///
				top20land Tlandhp SC ST OBC muslim childT west wealthy sibF ///
				sibM, stats(mean sd) columns(stats)	
				
est store A1

estpost tabstat Hindu1 Tiland_any Tiland age2 educ prop_sis educmhi educhi ///
				top20land Tlandhp SC ST OBC muslim childT west wealthy sibF ///
				sibM if femalec == 100, stats(mean sd) columns(stats)	
			 
est store B1

			
estpost tabstat Hindu1 Tiland_any Tiland age2 educ prop_sis educmhi educhi ///
				top20land Tlandhp SC ST OBC muslim childT west wealthy sibF ///
				sibM if femalec == 0, stats(mean sd) columns(stats)	
		 
est store C1	



* Output:
*--------
esttab A1 B1 C1 using "TableA6.rtf", replace ///
		title(Descriptive Statistics, Individuals) ///
		mtitle("All" "Women" "Men")  ///
		label cells(mean(fmt(2)) sd(fmt(2))) main(mean) aux(sd)


restore


	
			
		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

			
	

*-------------------------------------------------------------------------------
* Appendix Table A7: Descriptive Statistics, Dependent Variables
*-------------------------------------------------------------------------------


preserve

* set sample
*-----------
drop if (t >= 0 | birthyr < 1957)


estpost tabstat treat_age120 ///
		 vwreserv vwreserv3 dres1 da_diedpost da_postXdres1 ///	
		 dtreat_age120Xpost  ///
		 treat_age120Xdres1    ///
		 dtreat_age120XpostXdres1  ///
		 Tiland_any iland any_dowry_given attendance resolvesoc1, ///
		 stats(mean sd) columns(stats)	
		 
est store A1


estpost tabstat treat_age120 ///
		 vwreserv vwreserv3 dres1 da_diedpost da_postXdres1 ///	
		 dtreat_age120Xpost  ///
		 treat_age120Xdres1    ///
		 dtreat_age120XpostXdres1  ///
		 Tiland_any iland any_dowry_given attendance resolvesoc1 ///
		 if femalec == 100, ///
		 stats(mean sd) columns(stats)	
		 
est store B1

			
estpost tabstat treat_age120 ///
		 vwreserv vwreserv3 dres1 da_diedpost da_postXdres1 ///	
		 dtreat_age120Xpost  ///
		 treat_age120Xdres1    ///
		 dtreat_age120XpostXdres1  ///
		 Tiland_any iland attendance resolvesoc1 ///
		 if femalec == 0, ///
		 stats(mean sd) columns(stats)	
		 
est store C1	



* Output:
*--------

esttab A1 B1 C1 using "TableA7.rtf", replace ///
		title(Descriptive Statistics, Main Dependent and Independent Variables) ///
		mtitle("All" "Women" "Men")  ///
		label cells(mean(fmt(2)) sd(fmt(2))) main(mean) aux(sd)	
				

restore




			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
		
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

	


*-------------------------------------------------------------------------------
* Appendix Table A8: Descriptive Statistics: Villages without vs. with 
* 						reservations, excluding non-random reservations
*-------------------------------------------------------------------------------

	
preserve

* set appropriate sample
*-----------------------
keep if nr_short != 1


* descriptive statistics: by village averages
*--------------------------------------------	
collapse total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 vwreserv, ///
	by(villageid)


estpost tabstat  total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99, ///
		 stats (mean sem) columns (stats)	
		 
est store A

estpost tabstat  total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 ///
		 if vwreserv==1, ///
		 stats (mean sem) columns (stats)	
		 
est store B

			
estpost tabstat total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 ///
		 if vwreserv==0, ///
		 stats (mean sem) columns (stats)	

est store C		 
		
* ttests
		
estpost ttest total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99, ///
		 by(vwreserv) unequal
		
est store D


restore



esttab A B C using "TableA8_descriptive_statistics.rtf", label replace ///
		title(Descriptive statistics: Villages without vs. with reservations, ///
		excluding non-random implementers) ///
		mtitle("All Villages" "With Reservations" "Without Reservations") ///
		note("Source: Rural Economic and Demographic Survey, 2006" ///
		"Village level means are provided.") ///
		cells(mean(fmt(2))) main(mean) aux(sem)
		
		
esttab D using "TableA8_ttest.rtf", label replace main(b) ///
		mtitle("Difference of means: No Reservations - Reservations (t-score)") ///
		b(2) t(2) nostar
		
			

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
		
		


*-------------------------------------------------------------------------------
* Appendix Table A9: Descriptive Statistics: Villages without vs. with 
*						reservations, excluding non-random and late implementers
*-------------------------------------------------------------------------------

	
preserve

* set sample
*-----------
keep if (nr_short != 1 & late != 1)


* descriptive statistics: by village averages	
*--------------------------------------------
collapse total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 vwreserv, ///
	by(villageid)


estpost tabstat  total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99, ///
		 stats (mean sem) columns (stats)	
		 
est store A


estpost tabstat  total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 ///
		 if vwreserv==1, ///
		 stats (mean sem) columns (stats)	
		 
est store B

			
estpost tabstat total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 ///
		 if vwreserv==0, ///
		 stats (mean sem) columns (stats)	
		 
est store C	
		
* ttests
		
estpost ttest total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99, ///
		 by(vwreserv) 
		
est store D
restore


* output
*-------
esttab A B C using "TableA9_descriptive_statistics.rtf", label replace ///
		title(Descriptive statistics: Villages without vs. with reservations, ///
		excluding non-random and late implementers) ///
		mtitle("All Villages" "With Reservations" "Without Reservations") ///
		note("Source: Rural Economic and Demographic Survey, 2006" ///
		"Village level means are provided.") ///
		cells(mean(fmt(2))) main(mean) aux(sem)

		
		
esttab D using "TableA9_ttest.rtf", label replace main(b) ///
		mtitle("Difference of means: No Reservations - Reservations (t-score)") ///
		b(2) t(2) nostar
			
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**	
	


*-------------------------------------------------------------------------------
* Appendix Table A10: Descriptive Statistics: Villages without vs. with 
*						 reservations, Matched sample
*-------------------------------------------------------------------------------


	
preserve

* set sample: genetically matched villages
*-----------------------------------------
keep if genMatchId == 1


* descriptive statistics: by village averages	
*--------------------------------------------
collapse total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 vwreserv, ///
	by(villageid)

estpost tabstat  total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99, ///
		 stats (mean sem) columns (stats)	
		 
est store A


estpost tabstat  total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 ///
		 if vwreserv==1, ///
		 stats (mean sem) columns (stats)	
		 
est store B

			
estpost tabstat total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99 ///
		 if vwreserv==0, ///
		 stats (mean sem) columns (stats)	
		 
est store C	
		
* ttests
		
estpost ttest total_pop fem_ratio vpanch_pop1 ///
		 vpanch_memb1 vpscmemb_gp1 vpstmemb_gp1 vpobcmemb_gp1 ///
		 Vphind Vpmuslim Vpsc Vpst Vpobc vprmargin vprowner ///
		 vurlandpr vreslandpr vdrought_99 vflood_99 vpest_99, ///
		 by(vwreserv) 
		
est store D


label variable total_pop "Subdistrict population, 1991 census"
label variable fem_ratio "% women in the subdistrict (tehsil) population, 1991 census"
label variable vpanch_pop1 "Village population: first panchayat period"
label variable vpanch_memb1 "Number of panchayat members: first panchayat"
label variable vpscmemb_gp1 "% SC panchayat members: first panchayat period"
label variable vpstmemb_gp1 "% ST panchayat members: first panchayat period"
label variable vpobcmemb_gp1 "% OBC panchayat members: first panchayat period"
label variable Vphind "% Hindu in village population currently"
label variable Vpmuslim "% Muslim in village population currently"
label variable Vpsc "% SCs in village population currently"
label variable Vpst "% STs in village population currently"
label variable Vpobc "% OBCs in village population currently" 	
label variable vprmargin "% own less than 2 acres of land in village population currently"
label variable vprowner "% own land in village population currently"
label variable vurlandpr "Average price: unirrigated land currently (Rs.)"
label variable vreslandpr "Average price: residential land currently (Rs.)"
label variable vdrought_99 "% villages experienced drought, 1999"
label variable vflood_99 "% villages experienced flood, 1999"
label variable vpest_99 "% villages experienced pests, 1999"


* output
*-------
esttab A B C using "TableA10_descriptive_statistics.rtf", label replace ///
		title(Descriptive statistics: Villages without vs. with reservations, /// 
		matched sample) ///
		mtitle("All Villages" "With Reservations" "Without Reservations") ///
		note("Source: Rural Economic and Demographic Survey, 2006" ///
		"Village level means are provided.") ///
		cells(mean(fmt(2))) main(mean) aux(sem)
		
		
esttab D using "TableA10_ttest.rtf", label replace main(b) ///
		mtitle("Difference of means: No Reservations - Reservations (t-score)") ///
		b(2) t(2) nostar
		

restore



			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**	

			
			
			
*-------------------------------------------------------------------------------
* Appendix Table A.11: Reservations & Participation's Village-level Impact on 
*					   Female Inheritance
*-------------------------------------------------------------------------------


* Method I. Individual-level observations
*----------------------------------------
			
* STEP 1: Generate village-level averages of women's current GS participation
	
bysort villageid: egen attend_vil_fem = mean(attendance)	if fem == 1	

sum attend_vil_fem, de
*Note: mean = 0.097
			
bysort villageid: gen attend_Vil_fem_abovemean	= (attend_vil_fem > 0.097)
replace attend_Vil_fem_abovemean = . if (fem != 1)

label define attend_Vil_fem_abovemean 0 "Average - Below Average Participation" ///
								 1 "Above Average Participation"


* STEP 2: Generate village-level averages of women's inheritance

bysort villageid: egen Tiland_any_vil_fem = mean(Tiland_any) if fem == 1


* STEP 3: Descriptive statistics, individual-level

* Regardless of reservation status
correlate Tiland_any_vil_fem attend_Vil_fem_abovemean
* Note: correlation = 0.0668 

* In the presence of reservations for female pradhans
correlate Tiland_any_vil_fem attend_Vil_fem_abovemean if vwreserv3 == 1
* Note: correlation = 0.1097 


* STEP 4: Descriptive statistics WHERE RESERVATIONS, individual-level

estpost tabstat Tiland_any_vil_fem if vwreserv3 == 1,     ///
		by(attend_Vil_fem_abovemean) statistics(mean n sd)
		
est store AA
		

		
* Method II. Village-level observations
*--------------------------------------

* STEP 1: Create village-level means of female citizens
		
preserve

keep if fem == 1

collapse Tiland_any attendance vwreserv3, by(villageid)

egen attend_vf_mean = mean(attendance)

sum attend_vf_mean
*Note: mean = 0.1106328

gen attend_vf_abovemean = (attendance > 0.1106328)
label define attend_vf_abovemean 0 "Average - Below Average Participation" ///
								 1 "Above Average Participation"


* STEP 2: Descriptive statistics, village-level	
		
correlate Tiland_any attend_vf_abovemean if vwreserv3 == 1
* Note: correlation =  0.1637 


* STEP 3: Descriptive statistics WHERE RESERVATIONS, village-level

estpost tabstat Tiland_any if vwreserv3 == 1, by (attend_vf_abovemean) ///
		statistics(mean n sd)
		
est store BB	


* output, all
*------------
esttab AA BB using "TableA11.rtf", replace label ///
		title(Descriptive Statistics: Women's Mean Probability of Land ///
		Inheritance Conditional on Village Rate of Women's Gram Sabha ///
		Participation, Currently-Reserved Villages) ///
		mtitle("Individual Obs" "Village Obs") ///
		order (1 0) coeflabels (0 "Average - Below Average Participation" ///
								 1 "Above Average Participation")  ///
		note("Source: Rural Economic and Demographic Survey, 2006/9" ///
		"Village level means are provided.") ///
		cells(mean(fmt(3))) main(mean) aux(sem)
			
restore	

eststo clear


			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**	

		
			
			
*-------------------------------------------------------------------------------
* APPENDIX TABLE A12. Replicate main analysis from Goyal et al. 2013 
*-------------------------------------------------------------------------------


preserve

* set sample
*-----------
* For all birth years, for early reformer states excluding Kerala

keep if ((relation == 1 | relation == 8 | relation == 9) & ///
		 (stateid == 2 | stateid == 3 | stateid == 16 | stateid == 17))			


* Generating Fixed Effects & State trends: Goyal et al. 2013
*-----------------------------------------------------------

drop yrdum* stdum* sttrend*

ta stateid if stateid > 2, ge(stdum)

local uniq_stt = r(r)

forval i = 1/`uniq_stt' {
generate sttrend`i' = stdum`i'*birthyr
}


ta birthyr if birthyr < 2009, ge(yrdum)
local uniq_yr = r(r)


forval i = 1/`uniq_yr' {
generate fem_birthyr`i' = fem * yrdum`i'
}

foreach var of varlist sttrend* stdum* yrdum* fem_birthyr* {
replace `var' = 0 if `var' == . 
}

	  
*------------------------------------------------------------------------------
* Replicate Goyal et al. reform eligibility (father's death yr) for men & women
*------------------------------------------------------------------------------


* Generate father "treatment" by reform with linked father data 
*--------------------------------------------------------------

generate dtreatment_all = 0
replace dtreatment_all = 1 if ((stateid	==	16	&	da_deathyr	>=	1986)| ///
						  (stateid	==	17	&	da_deathyr	>=	1989)| ///
						  (stateid	==	2	&	da_deathyr	>=	1994)| ///
						  (stateid	==	3	&	da_deathyr	>=	1994))			  
						  
						  
replace dtreatment_all = . 	if (da_deathyr ==	.	&	da_live	==	0)
la var dtreatment_all "Father dies postreform (men+women)"


* Refine Father "treatment" by reform

generate dtreat_all = "." 		 if dtreatment_all == .
replace dtreat_all = "died post" if ((stateid == 16  & 	da_deathyr >= 1986) | ///
								(stateid == 17  &	da_deathyr >= 1989) | ///
								(stateid == 2   & 	da_deathyr >= 1994) | ///
								(stateid == 3   & 	da_deathyr >= 1994))

replace dtreat_all = "alive" if da_live == 1
replace dtreat_all = "died pre" if (dtreatment_all == 0  &  da_live != 1 & ///
													dtreat_all  != "died post")
replace dtreat_all = "." 	if (da_deathyr == . & da_live == 0)


* Generate Goyal et al's main treatment variable: Father's death post-reform

generate da_diedpost_all = (dtreat_all == "died post") 
la var da_diedpost_all "Father death post reform"


* Create Goyal et al's interaction terms 
*---------------------------------------

gen femXdtreatment_all = dtreatment_all * fem
la var femXdtreatment_all "Female*father dies post-HSAA"

gen femXda_diedpost_all = da_diedpost_all * fem
la var femXda_diedpost_all "Female*father's death post-HSAA"

gen da_diedpre6_all = (dtreat_all == "died pre" &  ///						  
				  (stateid	==	16	&	da_deathyr >= 1980 & da_deathyr	< 1986)| ///
				  (stateid	==	17	&	da_deathyr >= 1983 & da_deathyr < 1989)| ///
				  (stateid	==	2	&	da_deathyr >= 1988 & da_deathyr < 1994)| ///
				  (stateid	==	3	&	da_deathyr >= 1988 & da_deathyr < 1994))
				  
gen femXda_diedpre6_all = da_diedpre6_all * fem
la var femXda_diedpre6_all "Female* death pre-1-6 years"
gen da_diedpost5_all = (dtreat_all == "died post" & ///
				   (stateid	==	16	&	da_deathyr >= 1986 & da_deathyr < 1991)| ///
				   (stateid	==	17	&	da_deathyr >= 1989 & da_deathyr < 1994)| ///
				   (stateid	==	2	&	da_deathyr >= 1994 & da_deathyr < 1999)| ///
				   (stateid	==	3	&	da_deathyr >= 1994 & da_deathyr < 1999))	
				   
gen femXda_diedpost5_all = da_diedpost5_all * fem
la var femXda_diedpost5_all "Female* death post-0-5 years"			   
gen da_diedpost6_all = (dtreat_all == "died post" & ///
				   (stateid	==	16	&	da_deathyr >= 1991)| ///
				   (stateid	==	17	&	da_deathyr >= 1994)| ///
				   (stateid	==	2	&	da_deathyr >= 1999)| ///
				   (stateid	==	3	&	da_deathyr >= 1999))	
				   
gen femXda_diedpost6_all = da_diedpost6_all * fem
la var femXda_diedpost6_all "Female* death post-6+ years"		
		
	   			
* Generate Goyal et al's main regressions: T2 c1-2, 5
*--------------------------------------------------------			

eststo clear

eststo: quietly areg Tiland_any fem dtreatment_all femXdtreatment_all ///
				stdum* yrdum* fem_birthyr* if Relig == 1, ///
				absorb(interviewno) cl(villageid) 			
			
				
eststo: quietly areg Tiland_any fem da_diedpre6_all femXda_diedpre6_all ///
				da_diedpost5_all femXda_diedpost5_all ///
				da_diedpost6_all femXda_diedpost6_all ///
				stdum* yrdum* fem_birthyr* if Relig == 1, ///
                absorb(interviewno) cl(villageid) 		
											
			
eststo: quietly areg Tiland_any fem dtreatment_all femXdtreatment_all ///
				stdum* yrdum* fem_birthyr* if Relig != 1, ///
                absorb(interviewno) cl(villageid) 	


eststo: quietly areg Tiland_any fem da_diedpre6_all femXda_diedpre6_all ///
				da_diedpost5_all femXda_diedpost5_all ///
				da_diedpost6_all femXda_diedpost6_all ///
				stdum* yrdum* fem_birthyr* if Relig != 1, ///
                absorb(interviewno) cl(villageid) 		
					
restore


* output
*-------
esttab using "TableA12.rtf", b(2) se(2) replace label  ///
	   title(Reform's Short & Long-term Impact on Women's Land Inheritance, ///
	   Replication of Goyal et al. (2013)) ///
	   mtitle("Hindu" "Hindu"  "Non Hindu" "Non Hindu") ///
	   keep(fem femXdtreatment_all femXda_diedpre6_all ///
	   femXda_diedpost5_all femXda_diedpost6_all) ///
	   coeflabels(fem "Female" femXdtreatment_all "Female * father dies post-HSAA" ///
	   femXda_diedpre6_all "Female * father's death pre-1-6 years of reform" ///
	   femXda_diedpost5_all "Female* death post-0-5 years reform" ///
	   femXda_diedpost6_all "Female* death post-6+ years reform") ///
	   indicate("State FE = stdum*" "Birth year FE = yrdum*") ///
	   star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
	   varwidth(15) modelwidth(10) stats(r2 N,fmt(2 0) ///
	   label("R-sq" "N")) onecell
		
* Display p-values with coefficients for reference
							
esttab using "TableA12_ref.rtf", b(2) p(3) replace label  ///
	  title(Reform's Short & Long-term Impact on Women's Land Inheritance, ///
	  Replication of Goyal et al. (2013)) ///
	  mtitle("Hindu" "Hindu" "Non Hindu" "Non Hindu") ///
	  keep(fem femXdtreatment_all femXda_diedpre6_all ///
	  femXda_diedpost5_all femXda_diedpost6_all) ///
	  coeflabels(fem "Female" femXdtreatment_all "Female * father dies post-HSAA" ///
	  femXda_diedpre6_all "Female * father's death pre-1-6 years of reform" ///
	  femXda_diedpost5_all "Female* death post-0-5 years reform" ///
	  femXda_diedpost6_all "Female* death post-6+ years reform") ///
	  indicate("State FE = stdum*" "Birth year FE = yrdum*") ///
	  star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
	  varwidth(15) modelwidth(10) stats(r2 N,fmt(2 0) ///
	  label("R-sq" "N")) 
	
eststo clear
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**			

				
		
*-------------------------------------------------------------------------------
* APPENDIX TABLE A13. ROY REPLICATION
*-------------------------------------------------------------------------------

* Generate Roy's controls 
*------------------------

global controls_roy castedum_1 castedum_2 castedum_3 castedum_4 childT ///
					  wealthy							
	
		
* Generate Roy's measures of reform: grandfather & father treatments 
*-------------------------------------------------------------------

* Grandfather death specifications 
*---------------------------------
preserve

eststo clear


* Generate fixed effects for grandfather's year of death

tab grandp_deathyr, gen(gf_dyr)


* Replicate Roy's sample

keep if (stateid == 16 | stateid == 2 | stateid == 1 | stateid == 3 | ///
		stateid == 17)

keep if (fem == 1 & Hindu1 == 1 & landed == 1 & age2 >= 22 & mo_birthyr <= 1962)

keep if relation == 4


* Replicate Roy's analysis

* column 1: grandfather death, without controls, without grandfather death yr FE 

eststo: quietly reg Tiland_any treat_high treat_15 treat_20 ///
			g_diedpost gtreat_10Xpost gtreat_15Xpost gtreat_20Xpost ///
			stdum* yrdum* sttrend*, cl(stateid)
			
			
* column 2: grandfather death, with controls, without grandfather death yr FE

eststo: quietly reg Tiland_any treat_high treat_15 treat_20 ///
			g_diedpost gtreat_10Xpost gtreat_15Xpost gtreat_20Xpost ///
			$controls_roy stdum* yrdum* sttrend*, cl(stateid)	


*column 3: grandfather death, without controls, with grandfather death yr FE

eststo: quietly reg Tiland_any treat_high treat_15 treat_20 ///
			g_diedpost gtreat_10Xpost gtreat_15Xpost gtreat_20Xpost ///
			stdum* yrdum* sttrend* gf_dyr*, cl(stateid)
			
			
*column 4: grandfather death, with controls, with grandfather death yr FE			
			
eststo: quietly reg Tiland_any treat_high treat_15 treat_20 ///
			g_diedpost gtreat_10Xpost gtreat_15Xpost gtreat_20Xpost ///
			$controls_roy stdum* yrdum* sttrend* gf_dyr*, cl(stateid)			


restore 


* Father death specifications
*----------------------------

preserve


* Generate fixed effects for father's year of death

tab da_deathyr, gen(da_dyr)


* Replicate Roy's sample

keep if (stateid == 16 | stateid == 2 | stateid == 1 | stateid == 3 | ///
		stateid == 17) 

keep if (fem == 1 & Hindu1 == 1 & landed == 1 & age2 >= 22 & mo_birthyr <= 1962)

keep if (relation == 1 | relation == 9)


* Replicate Roy's analysis

*column 5: father death, without controls, without father death yr FE 

eststo: quietly reg Tiland_any treat_high treat_15 treat_20 ///
			da_diedpost dtreat_10Xpost dtreat_15Xpost dtreat_20Xpost ///
			stdum* yrdum* sttrend*, cl(stateid)

			
*column 6: father death, with controls, without father death yr FE 

eststo: quietly reg Tiland_any treat_high treat_15 treat_20 ///
			da_diedpost dtreat_10Xpost dtreat_15Xpost dtreat_20Xpost ///
			$controls_roy stdum* yrdum* sttrend*, cl(stateid)


*column 7: father death, without controls, with father death yr FE 

eststo: quietly reg Tiland_any treat_high treat_15 treat_20 ///
			da_diedpost dtreat_10Xpost dtreat_15Xpost dtreat_20Xpost ///
			stdum* yrdum* sttrend* da_dyr*, cl(stateid)
			
*column 8: father death, with controls, with father death yr FE 

eststo: quietly reg Tiland_any treat_high treat_15 treat_20 ///
			da_diedpost dtreat_10Xpost dtreat_15Xpost dtreat_20Xpost ///
			$controls_roy stdum* yrdum* sttrend* da_dyr*, cl(stateid)


restore


* ouput
*------
esttab using "TableA13.rtf", b(2) se(2) replace label ///
		title(Reform's Impact on Inheritance, Replication of Roy (2015)) ///
		keep(treat_high treat_15 treat_20 ///
		da_diedpost dtreat_10Xpost dtreat_15Xpost dtreat_20Xpost ///
		g_diedpost gtreat_10Xpost gtreat_15Xpost gtreat_20Xpost) ///
		order(treat_high treat_15 treat_20 ///
		g_diedpost gtreat_10Xpost gtreat_15Xpost gtreat_20Xpost ///
		da_diedpost dtreat_10Xpost dtreat_15Xpost dtreat_20Xpost) ///
		indicate("Controls = $controls_roy" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*" ///
		"Father's year of death FE = da_dyr*" ///
		"GF year of death FE = gf_dyr*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) 

eststo clear


			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**			

	

*-------------------------------------------------------------------------------
* Appendix Table A14. Reservations' Impact on Women's Inheritance, 
*							  Excluding Sisters without Brothers
*-------------------------------------------------------------------------------

* Generate new variable: any brothers
*------------------------------------
	
*-------------------------------------------------------------------------------
* R.0.1 Generate generation-specific indicators
*-------------------------------------------------------------------------------

gen g2 = (relation == 1 | relation == 8 | relation == 9)
bys interviewno: egen G2 = max(g2)

gen g3 = (relation == 4)
bys interviewno: egen G3 = max(g3)

gen g4 = (relation == 5)
bys interviewno: egen G4 = max(g4)


*-------------------------------------------------------------------------------
* R.0.2 Link generation-specific indicators to individual men
*-------------------------------------------------------------------------------

* create generation-specific indicators of brothers (# men in given family-gen.)
* note: want to know if a given woman of a given generation has *any* brothers.

generate g2_son = (femalec == 0 & g2 == 1)
bys interviewno: egen G2_sons=max(g2_son)

generate g3_son = (femalec == 0 & g3 == 1)
bys interviewno: egen G3_sons=max(g3_son)

generate g4_son = (femalec == 0 & g4 == 1)
bys interviewno: egen G4_sons=max(g4_son)

		
*-------------------------------------------------------------------------------
* R.0.3 Link generation-specific indicators to individual brothers
*-------------------------------------------------------------------------------
				
generate anybro = (g2 == 1 & G2_sons == 1 )

replace anybro 	= 1 if (g3 == 1 & G3_sons == 1 )	

replace anybro 	= 1 if (g4 == 1 & G4_sons == 1 )


*-------------------------------------------------------------------------------
* R.0.4 Run regressions excluding sisters without brothers
*-------------------------------------------------------------------------------

* Replication of Table 1, Columns 1-4
*------------------------------------

preserve 

eststo clear

* set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956 & anybro == 1)


* column 1 - no controls, state & year FE, state trends, landed Hindus only				
			
eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)


* column 2 - controls, state & year FE, state trends, landed Hindus only				
			
eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & anybro == 1), cl(villageid)
		

* column 3 - controls, state & yr FE & state trends, LH -non-random implementers

eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), ///
			cl(villageid)

			
* column 4 - controls, state & yr FE & state trends, LH-(non-random impl + late)

eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), ///
			cl(villageid)


restore

* Output
*-------		
esttab  using "TableA14_Column1to4.rtf", b(2) se(2) replace label ///
		title(Reservation's Impact on Women's Inheritance, Excluding Sisters ///
		without Brothers) mtitle("Target" "Target" "Target - NR" ///
		"Target-NR-late") keep(dres1 da_diedpost da_postXdres1) ///
		order(dres1 da_diedpost da_postXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell

esttab  using "TableA14_Column1to4_ref.rtf", b(2) p(3) replace label ///
		title(Reservation's Impact on Women's Inheritance, Excluding Sisters ///
		without Brothers) mtitle("Target" "Target" "Target - NR" ///
		"Target-NR-late") keep(dres1 da_diedpost da_postXdres1) ///
		order(dres1 da_diedpost da_postXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell		

eststo clear
		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

	
* Replication of Table 1, Columns 5-8
			
preserve 

eststo clear


* set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956 & anybro == 1)


* column 5 - only landed hindus, no controls, state & year FE, state trends, 
*			 excl. sisters w/o brothers			
			
eststo: quietly reg Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)

* column 6 - only landed hindus, controls, state & year FE, state trends, 
*			 excl. sisters w/o brothers			
			
eststo: quietly reg Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)

* column 7 - only landed hindus, controls, state & year FE, state tr - not rand, 
*			 excl. sisters w/o brothers			
			
eststo: quietly reg Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), ///
			cl(villageid)


* column 8 - only landed hindus, controls, state & year FE, state trends
*		     - not rand-late, excl. sisters w/o brothers			
			
eststo: quietly reg Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), ///
			cl(villageid)

est store F


* Output
*-------
esttab  using "TableA14_Column5to8.rtf", b(2) se(2) replace label ///
		title(Reservation's Impact on Women's Inheritance, Excluding Sisters ///
		without Brothers) mtitle("Target" "Target" "Target-NR" "Target-NR-late") ///
		keep(treat_age120 dres1 da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1 dtreat_age120XpostXdres1 ) ///
		order(dres1 da_diedpost da_postXdres1 treat_age120 dtreat_age120Xpost ///
			treat_age120Xdres1 dtreat_age120XpostXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell
		
esttab  using "TableA14_Column5to8_ref.rtf", b(2) p(3) replace label ///
		title(Reservation's Impact on Women's Inheritance, Excluding Sisters ///
		without Brothers) mtitle("Target" "Target" "Target-NR" "Target-NR-late") ///
		keep(treat_age120 dres1 da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1 dtreat_age120XpostXdres1 ) ///
		order(dres1 da_diedpost da_postXdres1 treat_age120 dtreat_age120Xpost ///
			treat_age120Xdres1 dtreat_age120XpostXdres1) ///
			star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell				
	
restore

eststo clear


			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
				
		
		
*-------------------------------------------------------------------------------
* APPENDIX TABLE A15: Reservations' Impact: Women's likelihood of Inheritance, 
* 					  Logit model
*-------------------------------------------------------------------------------

preserve 

eststo clear


*set sample
*----------

keep if (fem == 1 & t < 0 & birthyr > 1956)


* column 1 - no controls, state & year FE, state trends, all
			
eststo: quietly logit Tiland_any dres1 da_diedpost da_postXdres1 ///
			stdum* yrdum* sttrend*, cl(villageid)

* column 2 - controls, state & year FE, state trends, all		
			
eststo: quietly logit Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend*, cl(villageid)

* column 3 - controls, state & year FE & state trends, GenMatch Sample

eststo: quietly logit Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (genMatchId == 1), cl(villageid)

* column 4 - controls, state & year FE, state trends, landed Hindus only				
			
eststo: quietly logit Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)

* column 5 - controls, state & year FE & state trends, LH - non-random impl.

eststo: quietly logit Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), cl(villageid)

* column 6 - controls, state & year FE & state trends, LH - non-random impl.+late

eststo: quietly logit Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), ///
			cl(villageid)

restore


* Output
*-------		
esttab using "TableA15.rtf", b(2) se(2) replace label ///
		title(Reservation's Impact on Women's Likelihood of Inheritance; ///
		Logit Model) ///
		mtitle("All" "All" "Matched" "Target" "Target-NR" "Target-NR-late") ///
		keep(dres1 da_diedpost da_postXdres1) ///
		order(dres1 da_diedpost da_postXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
				star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(pr2 N,fmt(2 0) ///
		label("Pseudo R-sq" "N")) onecell
		
esttab using "TableA15_ref.rtf", b(2) p(3) replace label ///
		title(Reservation's Impact on Women's Likelihood of Inheritance; ///
		Logit Model) ///
		mtitle("All" "All" "Matched" "Target" "Target-NR" "Target-NR-late") ///
		keep(dres1 da_diedpost da_postXdres1) ///
		order(dres1 da_diedpost da_postXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
				star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(pr2 N,fmt(2 0) ///
		label("Pseudo R-sq" "N")) onecell

eststo clear
		

* Note: r2 doesn't appear in tables. Exclude "quietly" to see in STATA output.
			
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

			

*-------------------------------------------------------------------------------
* TABLE A16. Reservation's impact on daughters' inheritance 
*-------------------------------------------------------------------------------


preserve 

eststo clear

* set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956)


* column 1 - no controls, state & year FE, state trends, complete sample
eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			stdum* yrdum* sttrend*, cl(villageid)

* column 2 - controls, state & year FE, state trends, all		
eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend*, cl(villageid)

* column 3 - controls, state & year FE & state trends, GenMatch Sample
eststo: quietly reg Tiland_any dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (genMatchId == 1), cl(villageid)

restore
		
		
* Output
*-------
esttab  using "TableA16.rtf", b(2) se(2) replace label ///
		title(Reservation's Impact on Women's Likelihood of Inheritance, OLS) ///
		mtitle("All" "All" "Matched") ///
		keep(dres1 da_diedpost da_postXdres1) ///
		order(dres1 da_diedpost da_postXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell

esttab using "TableA16_ref.rtf", b(2) p(3) replace label ///
		title(Reservation's Impact on Women's Likelihood of Inheritance, OLS) ///
		mtitle("All" "All" "Matched") ///
		keep(dres1 da_diedpost da_postXdres1) ///
		order(dres1 da_diedpost da_postXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell			

		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			
			

			
*-------------------------------------------------------------------------------
* APPENDIX TABLE A17: Women's frequency of inheritance - Placebo 1984
*-------------------------------------------------------------------------------

* labeling variables
*-------------------

label variable dres_plac "Father died in 1984"


preserve 

eststo clear


* set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956)


* column 1 - no controls, state & year FE, state trends, complete sample
			
eststo: quietly reg Tiland_any dres_plac da_diedpost da_diedpostXdres_plac ///
			stdum* yrdum* sttrend*, cl(villageid)

* column 2 - controls, state & year FE, state trends, all		
			
eststo: quietly reg Tiland_any dres_plac da_diedpost da_diedpostXdres_plac ///
			$controls stdum* yrdum* sttrend*, cl(villageid)

* column 3 - controls, state & year FE & state trends, GenMatch Sample

eststo: quietly reg Tiland_any dres_plac da_diedpost da_diedpostXdres_plac ///
			$controls stdum* yrdum* sttrend* ///
			if (genMatchId == 1), cl(villageid)

* column 4 - controls, state & year FE, state trends, landed Hindus only				
			
eststo: quietly reg Tiland_any dres_plac da_diedpost da_diedpostXdres_plac ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)

* column 5 - controls, state & year FE & state trends, LH - non-random impl.

eststo: quietly reg Tiland_any dres_plac da_diedpost da_diedpostXdres_plac ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), cl(villageid)

* column 6 - controls, state & year FE & state trends, LH-(non-random impl+late)

eststo: quietly reg Tiland_any dres_plac da_diedpost da_diedpostXdres_plac ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), ///
			cl(villageid)

restore


* Output
*-------		
esttab  using "TableA17.rtf", b(2) se(2) replace label ///
		title(Placebo test: Reservation's Impact on Women's Likelihood of ///
		Inheritance) mtitle("All" "All" "Matched" "Target" "Target - NR" ///
		"Target-NR-late") keep(dres_plac da_diedpost da_diedpostXdres_plac) ///
		order(dres_plac da_diedpost da_diedpostXdres_plac) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell

esttab  using "TableA17_ref.rtf", b(2) p(3) replace label ///
		title(Placebo test: Reservation's Impact on Women's Likelihood of ///
		Inheritance) mtitle("All" "All" "Matched" "Target" "Target - NR" ///
		"Target-NR-late") keep(dres_plac da_diedpost da_diedpostXdres_plac) ///
		order(dres_plac da_diedpost da_diedpostXdres_plac) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell	

eststo clear
		
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			
		
		
*-------------------------------------------------------------------------------
* TABLE A18: Reservation's impact on daughters' area of land inheritance
*-------------------------------------------------------------------------------

preserve

eststo clear


* set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956)


* replace missing values of area of inheritance with zeroes

replace iland = 0 if iland ==.



* Create descriptive statistics for interpreting effect sizes
*------------------------------------------------------------

* women's mean ownership in landholding Hindu families

sum iland if (landed == 1 & Hindu1 == 1)
* Note: mean = .0304568, cited on pg. 21, including title and abstract pages.


* Regression analysis
*--------------------

* column 1 - no controls, state & year FE, state trends, landed Hindus only				
			
eststo: quietly reg iland dres1 da_diedpost da_postXdres1 ///
			stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)

* column 2 - controls, state & year FE, state trends, landed Hindus only				
			
eststo: quietly reg iland dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)


* column 3 - controls, state & year FE & state trends, Landed Hindu - NR impl.

eststo: quietly reg iland dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), cl(villageid)


* column 4 - controls, state & yr FE & state tr, Landed Hindu - NR & late impl.

eststo: quietly reg iland dres1 da_diedpost da_postXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), ///
			cl(villageid)

restore


* Output
*-------		
esttab  using "TableA18.rtf", b(2) se(2) replace ///
		label title(Reservation's Impact on Women's Area of Inheritance (acres)) ///
		mtitle("Target" "Target" "Target-NR" "Target-NR-late") ///
		keep(dres1 da_diedpost da_postXdres1) ///
		order(dres1 da_diedpost da_postXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell
		
esttab using "TableA18_ref.rtf", b(2) p(3) replace ///
		label title(Reservation's Impact on Women's Area of Inheritance (acres)) ///
		mtitle("Target" "Target" "Target-NR" "Target-NR-late") ///
		keep(dres1 da_diedpost da_postXdres1) ///
		order(dres1 da_diedpost da_postXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell

eststo clear
		

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			

			
*-------------------------------------------------------------------------------
* APPENDIX TABLE A.19: Reservation's Impact on Women's Likelihood of Inheritance, 
*						Conditional on Age & Father's death, Logit Model
*-------------------------------------------------------------------------------

preserve 

eststo clear


* set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956)


* column 1 - all no controls, state & year FE, state trends			
			
eststo: quietly logit Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			stdum* yrdum* sttrend*, cl(villageid)


* column 2 - all, controls, state & year FE, state trends		
			
eststo: quietly logit Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend*, cl(villageid)
			
* Note: for treat_age120Xdres1, standard errors vary by 0.0005 across 500 rounds
*		of regressions, ranging from 1.4948 to 1.4953 in the above specification
*		(which does not change the statistical significance of the term).


* column 3 - all: genetically matched, controls, state & year FE, state trends		
			
eststo: quietly logit Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* if genMatchId == 1, ///
			cl(villageid)


* column 4 - only landed hindus, controls, state & year FE, state trends			
			
eststo: quietly logit Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)
			

* column 5 - only landed hindus, controls, state & year FE, state tr - not rand			
			
eststo: quietly logit Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), cl(villageid)


* column 6 - only landed hindus, controls, state & year FE, state tr - not rand			
			
eststo: quietly logit Tiland_any treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), ///
			cl(villageid)

restore

* Output
*-------		
esttab using "TableA19.rtf", b(2) se(2) replace ///
		label title(Reservations' Dynamic Impact on Women's Inheritance, Logit ///
		Model) mtitle("All" "All" "Matched" "Target" "Target-NR" "Target-NR-late") ///
		keep(treat_age120 dres1 da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1 dtreat_age120XpostXdres1) ///
		order(treat_age120 dres1 da_diedpost da_postXdres1 dtreat_age120Xpost ///
			treat_age120Xdres1 dtreat_age120XpostXdres1) ///
			star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		varwidth(15) modelwidth(10) stats(pr2 N,fmt(2 0) ///
		label("Pseudo R-sq" "N")) onecell
		
esttab using "TableA19_ref.rtf", b(2) p(3) replace ///
		label title(Reservations' Dynamic Impact on Women's Inheritance, Logit ///
		Model) mtitle("All" "All" "Matched" "Target" "Target-NR" "Target-NR-late") ///
		keep(treat_age120 dres1 da_diedpost da_postXdres1 dtreat_age120Xpost ///
			treat_age120Xdres1 dtreat_age120XpostXdres1) ///
		order(treat_age120 dres1 da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1 dtreat_age120XpostXdres1 ) ///
			star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		varwidth(15) modelwidth(10) stats(pr2 N,fmt(2 0) ///
		label("Pseudo R-sq" "N")) onecell			
		
* Note: r2 doesn't appear in tables. Remove "quietly" to see in STATA output.
		
eststo clear		



			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

	
			
*---------------------------------------------------------------------
* TABLE A20 - RESERVATIONS - DYNAMIC TREATMENT FOR INHERITANCE & DOWRY
*---------------------------------------------------------------------
		
preserve 

eststo clear


* set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956)


**** Inheritance

* column 1 - all no controls, state & year FE, state trends			
			
eststo: quietly reg Tiland_any treat_age120  dres1 ///
			da_diedpost dtreat_age120Xpost  ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			stdum* yrdum* sttrend*, cl(villageid)

* column 2 - all, controls, state & year FE, state trends		
			
eststo: quietly reg Tiland_any treat_age120  dres1 ///
			da_diedpost dtreat_age120Xpost  ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend*, cl(villageid)

* column 3 - all: genetically matched, controls, state & year FE, state trends		
			
eststo: quietly reg Tiland_any treat_age120  dres1 ///
			da_diedpost dtreat_age120Xpost  ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* if (genMatchId == 1), ///
			cl(villageid)
			
***** Dowry			
			
* column 4 - all no controls, state & year FE, state trends			
			
eststo: quietly reg any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost  ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			stdum* yrdum* sttrend*, cl(villageid)

* column 5 - all, controls, state & year FE, state trends		
			
eststo: quietly reg any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost  ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend*, cl(villageid)

* column 6 - all: genetically matched, controls, state & year FE, state trends		
			
eststo: quietly reg any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost  ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1  ///
			$controls stdum* yrdum* sttrend* if (genMatchId == 1), ///
			cl(villageid)			
			
restore

* Output
*-------
esttab  using "TableA20.rtf", b(2) se(2) replace ///
		label title(Reservations' Dynamic Impact on Women's Inheritance and ///
		Dowry) mtitle("Inherit All" "Inherit All" "Inherit Matched" "Dowry All" ///
		"Dowry All" "Dowry Matched") keep(treat_age120 dres1 da_diedpost ///
		dtreat_age120Xpost da_postXdres1 treat_age120Xdres1 ///
		dtreat_age120XpostXdres1) ///
		order(treat_age120 dres1 da_diedpost da_postXdres1 dtreat_age120Xpost ///
			treat_age120Xdres1 dtreat_age120XpostXdres1 ) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell
		
esttab  using "TableA20_ref.rtf", b(2) p(3) replace ///
		label title(Reservations' Dynamic Impact on Women's Inheritance and ///
		Dowry) mtitle("Inherit All" "Inherit All" "Inherit Matched" "Dowry All" ///
		"Dowry All" "Dowry Matched") keep(treat_age120 dres1 da_diedpost ///
		dtreat_age120Xpost  ///
			da_postXdres1 treat_age120Xdres1 dtreat_age120XpostXdres1) ///
		order(treat_age120  dres1 da_diedpost dtreat_age120Xpost da_postXdres1 ///
			treat_age120Xdres1 dtreat_age120XpostXdres1) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell		
				

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**

			
			
*------------------------------------------------------------------------
* APPENDIX TABLE A21 - RESERVATIONS INCLUDED W/ALL FATHERS - Placebo 1984
*------------------------------------------------------------------------
		
preserve 

eststo clear


* set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956)


* column 1 - all no controls, state & year FE, state trends			
			
eststo: quietly reg Tiland_any treat_age120 dres_plac ///
			da_diedpost dtreat_age120Xpost  ///
			da_diedpostXdres_plac treat_age120Xdres_plac  ///
			dtreat_age120XpostXdres_plac  ///
			stdum* yrdum* sttrend*, cl(villageid)

* column 2 - all, controls, state & year FE, state trends		
			
eststo: quietly reg Tiland_any treat_age120 dres_plac ///
			da_diedpost dtreat_age120Xpost  ///
			da_diedpostXdres_plac treat_age120Xdres_plac  ///
			dtreat_age120XpostXdres_plac  ///
			$controls stdum* yrdum* sttrend*, cl(villageid)

* column 3 - all, genetically matched, controls, state & year FE, state trends		
			
eststo: quietly reg Tiland_any treat_age120 dres_plac ///
			da_diedpost dtreat_age120Xpost  ///
			da_diedpostXdres_plac treat_age120Xdres_plac  ///
			dtreat_age120XpostXdres_plac  ///
			$controls stdum* yrdum* sttrend* if genMatchId == 1, ///
			cl(villageid)

* column 4 - only landed hindus, controls, state & year FE, state trends			
			
eststo: quietly reg Tiland_any treat_age120 dres_plac ///
			da_diedpost dtreat_age120Xpost  ///
			da_diedpostXdres_plac treat_age120Xdres_plac  ///
			dtreat_age120XpostXdres_plac  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1== 1 & landed == 1), cl(villageid)

* column 5 - only landed hindus, controls, state & year FE, state tr - not rand,			
			
eststo: quietly reg Tiland_any treat_age120 dres_plac ///
			da_diedpost dtreat_age120Xpost  ///
			da_diedpostXdres_plac treat_age120Xdres_plac  ///
			dtreat_age120XpostXdres_plac  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), cl(villageid)

* column 6 - only landed hindus, controls, state & yr FE, state tr-not rand-late			
			
eststo: quietly reg Tiland_any treat_age120 dres_plac ///
			da_diedpost dtreat_age120Xpost  ///
			da_diedpostXdres_plac treat_age120Xdres_plac  ///
			dtreat_age120XpostXdres_plac  ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), cl(villageid)

restore

* Output
*-------		
esttab  using "TableA21.rtf", b(2) se(2) replace ///
		label title(Placebo test: Reservations' Dynamic Impact on Women's ///
		Inheritance) mtitle("All" "All" "Matched" "Target" "Target-NR" ///
		"Target-NR-late") keep(treat_age120 dres_plac da_diedpost ///
		    dtreat_age120Xpost da_diedpostXdres_plac treat_age120Xdres_plac ///
			dtreat_age120XpostXdres_plac) ///
		order(treat_age120 dres_plac da_diedpost da_diedpostXdres_plac ///
			 dtreat_age120Xpost treat_age120Xdres_plac  ///
			dtreat_age120XpostXdres_plac) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell
		
esttab using "TableA21_ref.rtf", b(2) p(3) replace ///
		label title(Placebo test: Reservations' Dynamic Impact on Women's ///
		Inheritance) mtitle("All" "All" "Matched" "Target" "Target-NR" ///
		"Target-NR-late") keep(treat_age120 dres_plac da_diedpost ///
		     dtreat_age120Xpost da_diedpostXdres_plac treat_age120Xdres_plac  ///
			 dtreat_age120XpostXdres_plac) ///
		order(treat_age120 dres_plac da_diedpost da_diedpostXdres_plac ///
			 dtreat_age120Xpost treat_age120Xdres_plac  ///
			 dtreat_age120XpostXdres_plac) ///
		indicate("Controls = $controls" "State FE = stdum*" ///
		"Birth year FE = yrdum*" "State trends = sttrend*") ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		varwidth(15) modelwidth(10) stats(r2_a N,fmt(2 0) ///
		label("Adj. R-sq" "N")) onecell		
		

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			

			
*-------------------------------------------------------------------------------
* Appendix Table A22. Genetically Matched Supplemental Tables
*-------------------------------------------------------------------------------		

preserve
		
char vwreserv [omit] 0
eststo clear		

keep if (reside == 1 & age_now > 17 & birthyr != .)			

* Col. 1. - Attendance, Genetically Matched, with controls, without trends
eststo: quietly xi: reg attendance fem vwreserv3 fem_vwreserv3 ///
		vdum* yrdum* $controls if (genMatchId == 1), cl(villageid)	
			
* Col 2: Effectiveness, Genetically matched - controls, no trends	
eststo: quietly xi: reg resolvesoc1 fem vwreserv3 fem_vwreserv3 ///
		vdum* yrdum* $controls if (genMatchId == 1), cl(villageid) 	


restore


esttab using "TableA22.rtf", b(2) se(2) replace label  ///
		title(Reservations' Impact: Attendance and Pradhan Effectiveness, OLS, ///
		Matched Samples) ///
		mtitle("Attendance, Matched" "Effectiveness, Matched") ///
		keep (fem vwreserv3 fem_vwreserv3) ///
		indicate("Controls = $controls" ///
		"Village FE = vdum*" "Birth year FE = yrdum*") ///
		varwidth(15) modelwidth(10) ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) 
		
esttab using "TableA22_ref.rtf", b(2) p(3) replace label  ///
		title(Reservations' Impact: Attendance and Pradhan Effectiveness, OLS, ///
		Matched Samples) ///
		mtitle("Attendance, Matched" "Effectiveness, Matched") ///
		keep (fem vwreserv3 fem_vwreserv3) ///
		indicate("Controls = $controls" ///
		"Village FE = vdum*" "Birth year FE = yrdum*") ///
		varwidth(15) modelwidth(10) ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) 		
		
* Note tables do not show adjusted r squared; to check remove "quietly" prefix 

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			

				
*-------------------------------------------------------------------------------
* APPENDIX TABLE A23: Reservations' Impact: Women's Participation in Gram Sabha, 
*					  Logit Model
*-------------------------------------------------------------------------------
	
			
preserve

* keep adult HH residents only (all survey respondents who can participate)
*--------------------------------------------------------------------------

keep if (reside == 1 & age_now > 17 & birthyr != .)

char fem [omit] 0

eststo clear				
		

* Col. 1. - All, without controls, without trends
eststo: quietly logit attendance fem vwreserv3  fem_vwreserv3  ///
		vdum* yrdum*, cl(villageid) 
		
* Col. 2. - All, controls, without trends
eststo: quietly logit attendance fem vwreserv3 fem_vwreserv3  ///
		vdum* yrdum* $controls, cl(villageid) 
		
* Col. 3. - All, controls, without trends, genetically matched sample
eststo: quietly logit attendance fem vwreserv3 fem_vwreserv3  ///
		vdum* yrdum* $controls if (genMatchId == 1), cl(villageid) 		
				
* Col. 4. - Excluding non-random, with controls, without trends		
eststo: quietly logit attendance fem vwreserv3 ///
		fem_vwreserv3 ///
		vdum* yrdum* $controls if (nr_short != 1), cl(villageid)
				
* Col. 5. - Excluding non-random+late, with controls, without trends		
eststo: quietly logit attendance fem vwreserv3 ///
		fem_vwreserv3 ///
		vdum* yrdum* $controls if (nr_short != 1 & late != 1), cl(villageid)		


restore

esttab using "TableA23.rtf", b(2) se(2) replace label ///
		title(Reservations' Impact on Women's Participation in Gram Sabha, ///
		Logit model) ///
		mtitle("All" "All" "Matched" "All-NR" "All-NR-late") ///
		keep ( fem vwreserv3 fem_vwreserv3) ///
		indicate("Controls = $controls" ///
		 "Village FE = vdum*" ///
		 "Birth year FE = yrdum*") ///
		varwidth(15) modelwidth(10) ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) onecell
		
esttab using "TableA23_ref.rtf", b(2) p(3) replace label ///
		title(Reservations' Impact on Women's Participation in Gram Sabha, ///
		Logit model) ///
		mtitle("All" "All" "Matched" "All-NR" "All-NR-late") ///
		keep ( fem vwreserv3 fem_vwreserv3) ///
		indicate("Controls = $controls" ///
		 "Village FE = vdum*" ///
		 "Birth year FE = yrdum*") ///
		varwidth(15) modelwidth(10) ///
		star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
		stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) onecell		

* Note tables do not show adjusted r squared; to check remove "quietly" prefix

	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			
		
		
*-------------------------------------------------------------------------------
* APPENDIX TABLE A24. RESERVATIONS,AGE,REFORM IMPACT: DOWRY (GIVEN), LOGIT MODEL
*-------------------------------------------------------------------------------
				
preserve 

eststo clear


* set sample
*-----------
keep if (fem == 1 & t < 0 & birthyr > 1956)


* column 1 - no controls, state & year FE, state trends, all			
			
eststo: quietly logit any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ///
			stdum* yrdum* sttrend*, cl(villageid)
	
* Note: for treat_age120Xdres1, standard errors vary by 0.01 across 500 rounds
*		of regressions, ranging from 0.51 to 0.52 in the above specification,
*		when estimating this specification to two decimal places (which does not 
*		change the statistical significance of the term). 
*		Note also that when estimating this term to four decimal places, the
*		standard errors consistently round to 0.52.

	
* column 2 - controls, state & year FE, state trends, all			
			
eststo: quietly logit any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ///
			$controls stdum* yrdum* sttrend*, cl(villageid)

			
* column 3 - controls, state & year FE, state trends, matched data			
			
eststo: quietly logit any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ///
			$controls stdum* yrdum* sttrend* if (genMatchId == 1), cl(villageid)
			

* column 4 - controls, state & year FE, state trends, landed Hindus only				
			
eststo: quietly logit any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1), cl(villageid)

* column 5 - C4 excluding non-random implementers

eststo: quietly logit any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1), cl(villageid)
			
			
* column 6 - C4 excluding non-random & late implementers

eststo: quietly logit any_dowry_given treat_age120 dres1 ///
			da_diedpost dtreat_age120Xpost ///
			da_postXdres1 treat_age120Xdres1  ///
			dtreat_age120XpostXdres1 ///
			$controls stdum* yrdum* sttrend* ///
			if (Hindu1 == 1 & landed == 1 & nr_short != 1 & late != 1), ///
			cl(villageid)		
			
restore

* Output
*-------
esttab using "TableA24.rtf", replace se title(Reservations' Dynamic Impact on ///
			Women's Dowry, Logit model) ///
			mtitle("All" "All" "Matched" "Target" "Target-NR" "Target-NR-late") ///
			cells(b(star fmt(2)) ///
			se(par fmt(2))) keep(treat_age120 dres1 da_diedpost ///
			dtreat_age120Xpost da_postXdres1 treat_age120Xdres1 ///
			dtreat_age120XpostXdres1) order(treat_age120 dres1 da_diedpost ///
			da_postXdres1 dtreat_age120Xpost treat_age120Xdres1 ///
			dtreat_age120XpostXdres1) ///
			indicate("Controls = $controls" ///
			"State FE = stdum*" "Cohort FE = yrdum*" "State trends = sttrend*") ///
			label 	star(+ 0.10 * 0.05 ** 0.01 *** 0.001) ///
			varwidth(15) modelwidth(10) ///
			stats(r2_a N, fmt(2 0) label("Adj. R-sq" "N")) onecell

eststo clear			
			

* Note tables do not show adjusted r squared; to check remove "quietly" prefix

			
			
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			
			
			
			
